多極子の等価演算子による表現

東北大理 楠瀬博明

Abstract

多極子交換模型の平均場や多極子の応答関数の計算に必要となる、多極子の等価演算子による表現の方法についてまとめる。球面調和関数の線形結合から直交位置座標の実関数多項式を構成し、多項式の各項を対応する全角運動量成分の対称和に置き換えることで等価演算子を導出する。また、導出を自動化するための Mathematica プログラムをあげる。最後に、有限磁場中での二次転移を決定するための平均場 (RPA) 感受率についてまとめておく。

1. 球面調和関数などの定義

本稿では、以下の定義を採用する。これらは、Hutchings による有名な解説 [1] や、応用群論 [2] で用いられているものと同一である。

- Legendre 多項式

$$P_n(x) = \frac{1}{2^n n!}\frac{d^n}{dx^n}(x^2-1)^n, \tag{1}$$

- Legendre 陪多項式

$$P_{nm}(x) = (1-x^2)^{m/2}\frac{d^m}{dx^m}P_n(x), \quad (-n \le m \le n), \quad [\text{Mathematica: } \times(-1)^m] \tag{2}$$

- 球面調和関数

$$\begin{aligned} Y_{\ell m}(\theta,\phi) &= (-1)^m\sqrt{\frac{2\ell+1}{4\pi}\frac{(\ell-m)!}{(\ell+m)!}}P_{\ell m}(\cos\theta)e^{im\phi}, \\ Y^*_{\ell m}(\theta,\phi) &= (-1)^m Y_{n-m}(\theta,\phi) \end{aligned} \tag{3}$$

2. 実表示 (Cartesian rep.)

球面調和関数は $m \neq 0$ のとき複素数なので、次の線形結合によって実関数を定義し直交座標で表す。

$$C_{\ell m} \equiv (-1)^m\left(\frac{4\pi}{2\ell+1}\right)^{1/2} r^\ell \frac{1}{\sqrt{2}}(Y_{\ell m}+Y^*_{\ell m}) = (-1)^m\sqrt{2}r^\ell\sqrt{\frac{(\ell-m)!}{(\ell+m)!}}P_{\ell m}\left(\frac{z}{r}\right)\cos(m\phi),$$

$$S_{\ell m} \equiv (-1)^m\left(\frac{4\pi}{2\ell+1}\right)^{1/2} r^\ell \frac{1}{\sqrt{2}i}(Y_{\ell m}-Y^*_{\ell m}) = (-1)^m\sqrt{2}r^\ell\sqrt{\frac{(\ell-m)!}{(\ell+m)!}}P_{\ell m}\left(\frac{z}{r}\right)\sin(m\phi),$$

$$J_{\ell m} \equiv \left(\frac{4\pi}{2\ell+1}\right)^{1/2} r^\ell Y_{\ell 0} = r^\ell P_\ell\left(\frac{z}{r}\right). \tag{4}$$

ここで、$\phi = \tan^{-1}(y/x)$, $r = \sqrt{x^2+y^2+z^2}$ であり、$C_{\ell m}, S_{\ell m}$ については $\ell \geq m \geq 1$ である。ここでの定義は、通常のものに比べて $\sqrt{4\pi/(2\ell+1)}$ 倍になっていることに注意する。これを用いて、球面調和関数の加法定理は

$$P_\ell(\hat{\boldsymbol{r}}\cdot\hat{\boldsymbol{r}}') = \frac{4\pi}{2\ell+1}\sum_{m=-\ell}^{\ell} Y^*_{\ell m}(\hat{\boldsymbol{r}})Y_{\ell m}(\hat{\boldsymbol{r}}') = \frac{1}{r^{2\ell}}\left[\sum_{m=1}^{\ell}\Big(C_{\ell m}(\hat{\boldsymbol{r}})C_{\ell m}(\hat{\boldsymbol{r}}') + S_{\ell m}(\hat{\boldsymbol{r}})S_{\ell m}(\hat{\boldsymbol{r}}')\Big) + J_{\ell 0}(\hat{\boldsymbol{r}})J_{\ell 0}(\hat{\boldsymbol{r}}')\right] \tag{5}$$

と書き換えることができる。点電荷模型の計算などでは、$\boldsymbol{r}$ を電子座標、$\boldsymbol{r}'$ を周りのイオンの座標ととる訳である。

$C_{\ell m}$ などの具体的な表式は、次の Mathematica プログラムで機械的に導出できる。 例えば、`TensorOp[2,{x,y,z}]` とすれば $C_{21}, C_{22}, S_{21}, S_{22}, J_{20}$ の順に結果が出力される。

sh.nb

```
(* spherical harmonics in Euclid coordinate *)
P[l_, m_, {r_, x_, y_, z_}] := Module[{a, d, ans},
  d = D[(a^2 - 1)^l, {a, l - m}];
  ans = (-1)^m((l + m)!/(l - m)!)/2^l/l!r^m(x^2 + y^2)^(-m/2) d;
  ans = ans /. a -> z/r;
  Simplify[ans]
];
Clm[l_, m_, {x_, y_, z_}] := Module[{r, f, ans, ph},
  f = TrigExpand[Cos[m ph]];
  f = f /. ph -> ArcCos[x(x^2 + y^2)^(-1/2)];
  ans = P[l, m, {r, x, y, z}]f r^l Sqrt[2]((l - m)!/(l + m)!)^(1/2);
  ans = ans /. r -> Sqrt[x^2 + y^2 + z^2];
  PowerExpand[Simplify[ans]]
];
Slm[l_, m_, {x_, y_, z_}] := Module[{r, f, ans, ph},
  f = TrigExpand[Sin[m ph]];
  f = f /. ph -> ArcSin[y(x^2 + y^2)^(-1/2)];
  ans = P[l, m, {r, x, y, z}]f r^l Sqrt[2] ((l - m)!/(l + m)!)^(1/2);
  ans = ans /. r -> Sqrt[x^2 + y^2 + z^2];
  PowerExpand[Simplify[ans]]
];
Jl0[l_, {x_, y_, z_}] := Module[{r, ans},
  ans = P[l, 0, {r, x, y, z}]r^l;
  ans = ans /. r -> Sqrt[x^2 + y^2 + z^2];
  Simplify[ans]
];
TensorOp[l_, r_] := Module[{m},
  Flatten[{Table[Clm[l, m, r], {m, l}],
    Table[Slm[l, m, r], {m, l}], Jl0[l, r]}]
```

```
];
```

$\ell \leq 6$ までの具体的な表式は以下の通りである。

s_tensor.nb

```
(* definition of real spherical tensors in cartesian coordinate *)
j00 = 1;
c11 = x;
s11 = y;
j10 = z;
c21 = Sqrt[3]xz;
c22 = (1/2)Sqrt[3](x^2-y^2);
s21 = Sqrt[3]yz;
s22 = Sqrt[3]xy;
j20 = (1/2)(-x^2-y^2+2z^2);
c31 = (-(1/2))Sqrt[3/2]x(x^2+y^2-4z^2);,
c32 = (1/2)Sqrt[15](x^2-y^2)z;
c33 = (1/2)Sqrt[5/2](x^3-3xy^2);
s31 = (-(1/2))Sqrt[3/2]y(x^2+y^2-4z^2);
s32 = Sqrt[15]xyz;
s33 = (1/2)Sqrt[5/2](3x^2y-y^3);
j30 = z^3-(3y^2z)/2-(3x^2z)/2;
c41 = (-(1/2))Sqrt[5/2]xz(3x^2+3y^2-4z^2);
c42 = (-(1/4))Sqrt[5](x^2-y^2)(x^2+y^2-6z^2);
c43 = (1/2)Sqrt[35/2](x^3-3xy^2)z;
c44 = (1/8)Sqrt[35](x^4-6y^2x^2+y^4);
s41 = (-(1/2))Sqrt[5/2]yz(3x^2+3y^2-4z^2);
s42 = (-(1/2))Sqrt[5]xy(x^2+y^2-6z^2);
s43 = (1/2)Sqrt[35/2](3x^2y-y^3)z;
s44 = (1/2)Sqrt[35]xy(x^2-y^2);
j40 = (1/8)(35z^4-30(x^2+y^2+z^2)z^2+3(x^2+y^2+z^2)^2);
c51 = (1/8)Sqrt[15]x(x^4+2(y^2-6z^2)x^2+y^4+8z^4-12y^2z^2);
c52 = (-(1/4))Sqrt[105](x^2-y^2)z(x^2+y^2-2z^2);
c53 = (-(1/8))Sqrt[35/2](x^3-3xy^2)(x^2+y^2-8z^2);
c54 = (3/8)Sqrt[35](x^4-6y^2x^2+y^4)z;
c55 = (3/8)Sqrt[7/2](x^5-10y^2x^3+5y^4x);
s51 = (1/8)Sqrt[15]y(x^4+2(y^2-6z^2)x^2+y^4+8z^4-12y^2z^2);
s52 = (-(1/2))Sqrt[105]xyz(x^2+y^2-2z^2);
s53 = (-(1/8))Sqrt[35/2](3x^2y-y^3)(x^2+y^2-8z^2);
s54 = (3/2)Sqrt[35]xy(x^2-y^2)z;
s55 = (3/8)Sqrt[7/2](y^5-10x^2y^3+5x^4y);
j50 = (1/8)(63z^5-70(x^2+y^2+z^2)z^3+15(x^2+y^2+z^2)^2z);
c61 = (1/8)Sqrt[21]xz(5x^4+10(y^2-2z^2)x^2+5y^4+8z^4-20y^2z^2);
c62 = (1/16)Sqrt[105/2](x^2-y^2)(x^4+2(y^2-8z^2)x^2+y^4+16z^4
     -16y^2z^2);
```

```
c63 = (-(1/8))Sqrt[105/2](x^3-3xy^2)z(3x^2+3y^2-8z^2);
c64 = (-(3/16))Sqrt[7](x^4-6y^2x^2+y^4)(x^2+y^2-10z^2);
c65 = (3/8)Sqrt[77/2](x^5-10y^2x^3+5y^4x)z;
c66 = (1/16)Sqrt[231/2](x^6-15y^2x^4+15y^4x^2-y^6);
s61 = (1/8)Sqrt[21]yz(5x^4+10(y^2-2z^2)x^2+5y^4+8z^4-20y^2z^2);
s62 = (1/8)Sqrt[105/2]xy(x^4+2(y^2-8z^2)x^2+y^4+16z^4-16y^2z^2);
s63 = (-(1/8))Sqrt[105/2](3x^2y-y^3)z(3x^2+3y^2-8z^2);
s64 = (-(3/4))Sqrt[7]xy(x^2-y^2)(x^2+y^2-10z^2);
s65 = (3/8)Sqrt[77/2](y^5-10x^2y^3+5x^4y)z;
s66 = (1/8)Sqrt[231/2]xy(3x^4-10y^2x^2+3y^4);
j60 = (1/16)(231z^6-315(x^2+y^2+z^2)z^4+105(x^2+y^2+z^2)^2z^2
      -5(x^2+y^2+z^2)^3);
```

3. 立方調和関数

点群 O_h の既約表現に属する基底の表現として立方調和関数を考える。これらは、前節の実関数球面調和関数の線形結合として与えられる。具体的な定義は以下の通り。

c_tensor.nb

```
(* definition of irrducible tensors under cubic symmetry *)
(* rank, irreducible rep., multiplicity, component *)
f0111 = j00;
f1411 = c11;
f1412 = s11;
f1413 = j10;
f2311 = j20;
f2312 = c22;
f2511 = s21;
f2512 = c21;
f2513 = s22;
f3211 = s32;
f3411 = 1/Sqrt[8](Sqrt[5]c33 - Sqrt[3]c31);
f3412 = -1/Sqrt[8](Sqrt[5]s33 + Sqrt[3]s31);
f3413 = j30;
f3511 = -1/Sqrt[8](Sqrt[3]c33 + Sqrt[5]c31);
f3512 = 1/Sqrt[8](-Sqrt[3]s33 + Sqrt[5]s31);
f3513 = c32;
f4111 = 1/Sqrt[12](Sqrt[5]c44 + Sqrt[7]j40);
f4311 = -1/Sqrt[12](Sqrt[7]c44 - Sqrt[5]j40);
f4312 = -c42;
```

```
f4411 = -1/Sqrt[8](s43 + Sqrt[7]s41);
f4412 = -1/Sqrt[8](c43 - Sqrt[7]c41);
f4413 = s44;
f4511 = 1/Sqrt[8](Sqrt[7]s43 - s41);
f4512 = -1/Sqrt[8](Sqrt[7]c43 + c41);
f4513 = s42;
f5311 = s54;
f5312 = -s52;
f5411 = 1/8/Sqrt[2](3Sqrt[7]c55 - Sqrt[35]c53 + Sqrt[30]c51);
f5412 = 1/8/Sqrt[2](3Sqrt[7]s55 + Sqrt[35]s53 + Sqrt[30]s51);
f5413 = j50;
f5421 = 1/16(Sqrt[10]c55 + 9Sqrt[2]c53 + 2Sqrt[21]c51);
f5422 = 1/16(Sqrt[10]s55 - 9Sqrt[2]s53 + 2Sqrt[21]s51);
f5423 = c54;
f5511 = 1/4/Sqrt[2](-Sqrt[15]c55 - Sqrt[3]c53 + Sqrt[14]c51);
f5512 = 1/4/Sqrt[2](Sqrt[15]s55 - Sqrt[3]s53 - Sqrt[14]s51);
f5513 = c52;
f6111 = 1/Sqrt[8](-Sqrt[7]c64 + j60);
f6211 = 1/4(-Sqrt[5]c66 + Sqrt[11]c62);
f6311 = 1/Sqrt[8](c64 + Sqrt[7]j60);
f6312 = 1/4(Sqrt[11]c66 + Sqrt[5]c62);
f6411 = 1/8(-Sqrt[22]s65 - Sqrt[30]s63 + 2Sqrt[3]s61);
f6412 = 1/8(Sqrt[22]c65 - Sqrt[30]c63 - 2Sqrt[3]c61);
f6413 = s64;
f6511 = 1/16(Sqrt[3]s65 + Sqrt[55]s63 + 3Sqrt[22]s61);
f6512 = 1/16(Sqrt[3]c65 - Sqrt[55]c63 + 3Sqrt[22]c61);
f6513 = s66;
f6521 = 1/16(Sqrt[165]s65 - 9s63 + Sqrt[10]s61);
f6522 = 1/16(Sqrt[165]c65 + 9c63 + Sqrt[10]c61);
f6523 = s62;
```

4. 等価演算子 (Stevens' operator method)

Wigner-Eckart の定理 [2,3] によれば、全角運動量 J が一定の状態の間では、x, y, z の多項式の行列要素は、対応する全角運動量演算子の多項式の対称和の行列要素に比例する。ただし、演算子 J は時間反転に対して奇の性質があるため、同じ対称性を持つ ℓ が奇数の磁気 2^ℓ 極子、または、ℓ が偶数の電気 2^ℓ 極子だけが、等価演算子の方法を用いて表すことができる。以上から具体的な処方箋は、さきほど求めた実多項式において、

$$x^\ell y^m z^n \to \frac{\ell!m!n!}{(\ell+m+n)!}\sum_{\mathcal{P}}\mathcal{P}(\hat{J}_x^\ell \hat{J}_y^m \hat{J}_z^n) \tag{6}$$

の置き換えをすればよい。ここで、和は $J_x\ J_y, J_z$ に関してあらゆる置換を行ったものについてとる。このような置き換えによって得られた等価演算子は Hermite であり、その固有値は実数である。

上記の等価演算子 ($\hat{A}$ をつけて表す) と、いわゆる Stevens 演算子 [1,3] との関係は以下の通りである。

$$O_2{}^0 = 2\hat{J}_{20},\ \ O_2{}^2 = \frac{2}{\sqrt{3}}\hat{C}_{22},$$
$$O_4{}^0 = 8\hat{J}_{40},\ \ O_4{}^2 = \frac{4}{\sqrt{5}}\hat{C}_{42},\ \ O_4{}^3 = \frac{4}{\sqrt{70}}\hat{C}_{43},\ \ O_4{}^3(s) = \frac{4}{\sqrt{70}}\hat{S}_{43},$$
$$O_4{}^4 = \frac{8}{\sqrt{35}}\hat{C}_{44},\ \ O_4{}^4(s) = \frac{8}{\sqrt{35}}\hat{S}_{44},$$
$$O_6{}^0 = 16\hat{J}_{60},\ \ O_6{}^2 = \frac{32}{\sqrt{210}}\hat{C}_{62},\ \ O_6{}^3 = \frac{16}{\sqrt{210}}\hat{C}_{63},\ \ O_6{}^4 = \frac{16}{3\sqrt{7}}\hat{C}_{64},\ \ O_6{}^6 = \frac{32}{\sqrt{462}}\hat{C}_{66} \tag{7}$$

等価演算子を機械的に求める Mathematica プログラムを以下にあげる。最初に `J=MakeJvec[4]` と呼び出して、$J = 4$ の全角運動量演算子の各成分の行列要素 (9×9, J_z が減少する順) を変数 `J` に代入する。次に置換したい多項式、例えば `f2311` を使って、`ReplaceOperator[f2311,J]` と呼び出せば、9×9 の等価演算子行列が得られる。`f2311` の代わりに式 (例えば `ReplaceOperator[f2311+f2312,J]`) などを渡すこともできる。

replace_op.nb

```
(* Create angular momentum matrices *)
CreateJMatrix[J_] := Module[{Jp, Jm, Jz, Jvec, i},
   Jvec = Table[i, {i, J, -J, -1}]; (* sequence of basis *)
   Del[x_, y_] := Switch[x, y, 1, _, 0];(* Kronecker's delta *)
   Jp[j1_, j2_] := Del[j1, j2 + 1]Sqrt[(J - j2)(J + j2 + 1)];
   Jm[j1_, j2_] := Del[j1, j2 - 1]Sqrt[(J + j2)(J - j2 + 1)];
   Jz[j1_, j2_] := Del[j1, j2]j2;
   List[
      (Outer[Jp, Jvec, Jvec] + Outer[Jm, Jvec, Jvec])/2,
      (Outer[Jp, Jvec, Jvec] - Outer[Jm, Jvec, Jvec])/(2I),
      (Outer[Jz, Jvec, Jvec])
   ]
];
(* Replace an expression as a sum of permutations *)
ReplaceOperator[ex_, j_] := Module[{Lbar, GetExp, ReplaceTerm, term,
   jx, jy, jz, f},
   Lbar[L_] := Apply[Plus, Apply[Dot, Permutations[L], 1]];
   GetExp[c_?NumericQ val_] := {c, Exponent[val, x], Exponent[val, y],
      Exponent[val, z]};
   GetExp[val_] := {1, Exponent[val, x], Exponent[val, y],
      Exponent[val, z]};
   ReplaceTerm[et_] := Module[{c, l, m, n, ls},
      {c, l, m, n} = GetExp[et];
      ls = Join[Table[jx, {l}], Table[jy, {m}], Table[jz, {n}]];
      c l!m!n!/(l + m + n)!Lbar[ls]
```

```
  ];
  f[x_] := ReplaceTerm[x];
  f[x_ + y_] := ReplaceTerm[x] + f[y];
  Simplify[f[Expand[ex]] /. {jx -> j[[1]], jy -> j[[2]], jz -> j[[3]]}]
];
```

5. 平均場 (RPA) 感受率

自発的な秩序のない相からの二次転移点を平均場で求めるには RPA 感受率が発散する温度 T を求めればよい。サイト i での多極子演算子 α を X_i^α で表し、以下の一般的なハミルトニアンを考える。

$$H = \sum_i H_{\mathrm{CEF}}^{(i)} - \sum_{i\alpha} X_i^\alpha h^\alpha - \frac{1}{2}\sum_{ij}\sum_{\alpha\beta} D_{ij}^{\alpha\beta} X_i^\alpha X_j^\beta - \sum_{i\alpha} X_i^\alpha \phi_i^\alpha \tag{8}$$

第一項は結晶場ハミルトニアン、第二項は Zeeman 項 $(-g_J\mu_{\mathrm{B}}\sum_i \boldsymbol{J}_i\cdot\boldsymbol{H})$ である。第三項はサイト間の交換相互作用を表す。最後の項は、感受率を計算するために導入した多極子 α に共役な場である。有限磁場中で、微小な場 ϕ_i^α がかかったときの多極子の熱平均を $\langle X_i^\alpha\rangle = \langle X_i^\alpha\rangle_0 + \langle X_i^\alpha\rangle_\phi$ と書く。ここで、$\langle X_i^\alpha\rangle_\phi$ は、ϕ_i^α によって誘起された ϕ に比例する量とする。有限磁場下では $\langle X_i^\alpha\rangle_0$ はゼロでないものもあることに注意。

平均場ハミルトニアンは、以下のようになる。

$$H_{\mathrm{MF}} = \sum_i \left[H_0 - \sum_\alpha X_i^\alpha \lambda_i^\alpha \right], \tag{9}$$

$$H_0 = H_{\mathrm{CEF}}^{(i)} - \sum_\alpha \left(h^\alpha + \sum_{j\beta} D_{ij}^{\alpha\beta}\langle X_j^\beta\rangle_0 \right) X_i^\alpha, \quad \lambda_i^\alpha = \phi_i^\alpha + \sum_{j\beta} D_{ij}^{\alpha\beta}\langle X_j^\beta\rangle_\phi$$

自発的な秩序のない場合を考えているので、H_0 はサイトによらない。H_0 のエネルギーと固有状態を $H_0|m\rangle = E_m|m\rangle$ と表すと、

$$\langle X_i^\alpha\rangle_0 = \sum_m f_m\langle m|X_i^\alpha|m\rangle, \quad f_m = \frac{e^{-\beta E_m}}{\sum_m e^{-\beta E_m}} \tag{10}$$

であり、$E_m, |m\rangle$ が $\langle X_i^\alpha\rangle_0$ に暗に依存しているため、空間的に一様な $\langle X_i^\alpha\rangle_0$ を求めるセルフ・コンシステント方程式である。系 H_0 の局所感受率は、線形応答理論より $x_{nm} = (E_n - E_m)/T$ として

$$T\chi_0^{\alpha\beta} = T\int_0^\beta d\lambda \langle (X_i^\alpha(\lambda) - \langle X_i^\alpha\rangle_0)(X_i^\beta - \langle X_i^\beta\rangle_0)\rangle_0 = \sum_{mn} f_m \frac{1-e^{-x_{nm}}}{x_{nm}}\langle m|X_i^\alpha|n\rangle\langle n|X_i^\beta|m\rangle - \langle X_i^\alpha\rangle_0\langle X_i^\beta\rangle_0 \tag{11}$$

と求まる。次に、ϕ_i^α 場を考慮すると

$$\langle X_i^\alpha \rangle_\phi = \sum_\beta \chi_0^{\alpha\beta} \lambda_i^\beta = \sum_\gamma \chi_0^{\alpha\gamma} \left[\phi_i^\gamma + \sum_{k\delta} D_{ik}^{\gamma\delta} \langle X_k^\delta \rangle_\phi \right] \tag{12}$$

となる。求めたい RPA 感受率は、$\chi_{ij}^{\alpha\beta} = \partial \langle X_i^\alpha \rangle_\phi / \partial \phi_j^\beta |_{\phi=0}$ である。上の式の両辺を ϕ_j^β で微分して

$$\chi_{ij}^{\alpha\beta} = \sum_\gamma \chi_0^{\alpha\gamma} (\delta_{ij}\delta_{\gamma\beta} + \sum_{k\delta} D_{ik}^{\gamma\delta} \chi_{kj}^{\delta\beta}) \tag{13}$$

を得る。この式を Fourier 変換して

$$\chi^{\alpha\beta}(\boldsymbol{q}) = \sum_\gamma [\hat{1} - \hat{\chi}_0 \cdot \hat{D}(\boldsymbol{q})]^{-1}_{\alpha\gamma} \chi_0^{\gamma\beta} \tag{14}$$

となる。ここで、$D^{\alpha\beta}(\boldsymbol{q}) = \sum_r e^{-i\boldsymbol{q}\cdot\boldsymbol{\eta}_r} D^{\alpha\beta}(\boldsymbol{\eta}_r)$ である。特に、最近接サイトにのみ相互作用 $D^{\alpha\beta}$ が働く場合、$D^{\alpha\beta}(\boldsymbol{q}) = zD^{\alpha\beta}\gamma_{\boldsymbol{q}}$ (z は最近接サイト数) となる。$\gamma_{\boldsymbol{q}} = (1/z)\sum_r e^{-i\boldsymbol{q}\cdot\boldsymbol{\eta}_r}$ は格子の形状に依存する。代表的な格子については以下の通り。

- 一次元格子: $\gamma_{\boldsymbol{q}} = \cos(qa)$
- 二次元正方格子: $\gamma_{\boldsymbol{q}} = \dfrac{1}{2}[\cos(q_x a) + \cos(q_y a)]$
- 三次元単純格子: $\gamma_{\boldsymbol{q}} = \dfrac{1}{3}[\cos(q_x a) + \cos(q_y a) + \cos(q_z a)]$
- bcc: $\gamma_{\boldsymbol{q}} = \cos(q_x a/2)\cos(q_y a/2)\cos(q_z a/2)$
- fcc: $\gamma_{\boldsymbol{q}} = \dfrac{1}{3}[\cos(q_x a/2)\cos(q_y a/2) + \cos(q_y a/2)\cos(q_z a/2) + \cos(q_z a/2)\cos(q_x a/2)]$

二次転移は

$$\det \left[\delta_{\alpha\beta} - \sum_\gamma \chi_0^{\alpha\gamma} D^{\gamma\beta}(\boldsymbol{q}) \right] = 0 \tag{15}$$

を満たす中で、最大の T_c を持つ波数 $\boldsymbol{q} = \boldsymbol{Q}$ で起こり、その秩序変数は、上式括弧内 $T = T_c$、$\boldsymbol{Q}$ での行列の固有ベクトルとして求まる。

References

[1] M.T. Hutchings, Solid State Phys. **16** (1964) 227.

[2] 小野寺、田辺、犬井著「応用群論」(裳華房 1976).

[3] K.W.H. Stevens, Proc. Phys. Soc. A**65** (1952) 209.